ACADEMIC PRESS

J. Math. Anal. Appl. 269 (2002) 317–331

Journal of MATHEMATICAL ANALYSIS AND APPLICATIONS

www.academicpress.com

Symbolic evaluation of coefficients in Airy-type asymptotic expansions

Raimundas Vidunas [a,b] and Nico M. Temme [b,*]

[a] *Korteweg de Vries Instituut voor Wiskunde, University of Amsterdam, Plantage Muidergracht 24, 1018 TV Amsterdam, The Netherlands*

[b] *CWI, P.O. Box 94079, 1090 GB Amsterdam, The Netherlands*

Received 26 February 2001; received in revised form 29 October 2001; accepted 16 November 2001

Submitted by B.C. Berndt

Abstract

Computer algebra algorithms are developed for evaluating the coefficients in Airy-type asymptotic expansions that are obtained from integrals with a large parameter. The coefficients are defined from recursive schemes obtained from integration by parts. An application is given for the Weber parabolic cylinder function. © 2002 Elsevier Science (USA). All rights reserved.

Keywords: Airy-type asymptotic expansions; Maple algorithms; Parabolic cylinder functions

1. Introduction

When constructing uniform asymptotic expansions of solutions of differential equations or of functions defined by integrals, usually a difficulty arises when the coefficients of the expansion are constructed. As shown in [1] for the Airy-type expansions of Bessel functions, recursion relations for the coefficients can be obtained for the case that the expansion is obtained by using a linear second order differential equation.

* Corresponding author.

E-mail addresses: vidunas@wins.uva.nl, vidunas@cwi.nl (R. Vidunas), nicot@cwi.nl (N.M. Temme).

0022-247X/02/$ – see front matter
PII: S0022-247X(02)00026-4

In many publications this method has been used (for example, in [2] and [3]), and for expansions involving Bessel functions or parabolic cylinder functions similar results are available. Having such a recursion relation for the coefficients does not always give the possibility to obtain analytic expressions of a number of coefficients, because the recursion involves integrals of previous coefficients together with a function that is not easy to handle. Sometimes the coefficients can be explicitly expressed in terms of coefficients of simpler expansions because different types of expansions may be valid in overlapping domains. See for the Bessel functions the relations in [1, p. 425, Exercise 10.3] or [4, p. 368, formula (9.3.40)].

For special functions usually the same type of uniform expansions can be obtained by using integral representations of the functions. Sometimes, in a particular problem, the integral is the only tool available for constructing uniform expansions. By using transformations of variables in the integrals, these representations can be transformed into standard forms for which an integration by parts procedure can be used to obtain expansions in terms of, for example, Airy functions.

Although it is usually not possible to derive recursion relations for the coefficients obtained in this way, in all cases for special functions known so far, it is possible to construct a number of coefficients, and only because of the complexity of the problem, which implies limitations with respect to available computer memory when doing symbolic computations, there is an upper bound for this number. An advantage of the differential equation approach is the possibility to construct realistic and sharp error bounds for the remainders in the expansions; it is not known how to obtain similar bounds in the approach based on integral representations.

In this paper we use integral representations and give Maple algorithms for constructing the coefficients in uniform asymptotic expansions involving Airy functions. First we describe how to obtain the coefficients for a general case. For an application we obtain the coefficients for the case of a special function called parabolic cylinder function. Straightforward computations are often complicated by appearance of algebraic roots in the output or intermediate expressions. These algebraic roots can be avoided by replacing some parameters with algebraic expressions in suitable new variables. In the example of parabolic cylinder function we avoid computations with algebraic roots by using variable u (instead of t), and at the end we simplify the output by introducing variable ξ. In the final section we give the Maple code used for this example.

2. Airy-type asymptotic expansions

We consider integrals of the form

$$F_\eta(z) = \frac{1}{2\pi i}\int_C e^{z[(1/3)t^3-\eta t]} f(t)\,dt, \tag{1}$$

where the contour starts at infinity with $\operatorname{ph} t = -\pi/3$ and returns to infinity with $\operatorname{ph} t = \pi/3$. We assume that the function $f(t)$ is analytic in the neighbourhood of the contour; z and η are complex parameters, z is large.

In the case $f(t) = 1$ we obtain the Airy function [5, p. 101]

$$\frac{1}{2\pi i}\int_C e^{z[(1/3)t^3-\eta t]}\,dt = z^{-1/3}\,\mathrm{Ai}(\eta z^{2/3}). \tag{2}$$

For more general functions f the asymptotic expansion of $F_\eta(z)$ can be given in terms of this Airy function. The asymptotic feature of this type of integral is that the phase function $\phi(t) = (1/3)t^3 - \eta t$ has two saddle points at $\pm\sqrt{\eta}$ that coalesce when $\eta \to 0$, and it is not possible to describe the asymptotic behaviour of $F_\eta(z)$ in terms of simple functions when η is small. When the parameter η is positive and bounded away from 0, one can perform a saddle point analysis on (1) and use a conformal mapping $\phi(t) - \phi(\sqrt{\eta}) = (1/2)u^2$ with the condition $u(\sqrt{\eta}) = 0$. We obtain

$$F_\eta(z) = \frac{e^{z\phi(\sqrt{\eta})}}{2\pi i}\int_{-i\infty}^{i\infty} e^{(1/2)zu^2} g(u)\,du,$$

where $g(u) = f(t)\,dt/du$, with $dt/du = u/(t^2-\eta)$, which is regular at the positive saddle point, but not at the negative saddle point. It follows that, when η becomes small, a singularity due to dt/du in the u-plane approaches the origin, and an expansion of dt/du at $u = 0$ will have coefficients that become infinite as $\eta \to 0$. Hence, by using the standard saddle point method we obtain an expansion that is not uniformly valid as $\eta \to 0$.

A modification of the saddle point method is possible by taking into account both saddle points. We give an integration by parts procedure that is a variant of Bleistein's method introduced in [6] (for a different class of integrals), and that gives the requested uniform expansion.

We assume that f is an analytic function in a certain domain G and write

$$f(t) = \alpha_0 + \beta_0 t + (t^2-\eta)g(t), \tag{3}$$

where

$$\alpha_0 = \frac{1}{2}[f(\sqrt{\eta}) + f(-\sqrt{\eta})],$$
$$\beta_0 = \frac{1}{2\sqrt{\eta}}[f(\sqrt{\eta}) - f(-\sqrt{-\eta})]. \tag{4}$$

Clearly $\alpha_0 \to f(0)$, $\beta_0 \to f'(0)$ as $\eta \to 0$. We have the Cauchy integral representation

$$f(t) = \frac{1}{2\pi i} \oint_{\{t\}} \frac{f(s)}{s-t}\, ds,$$

where the contour encircles the point t in the anti-clockwise direction. Similarly,

$$\alpha_0 = \frac{1}{2}\left[\frac{1}{2\pi i} \oint_{\{\sqrt{\eta}\}} \frac{f(s)}{s-\sqrt{\eta}}\, ds + \frac{1}{2\pi i} \oint_{\{-\sqrt{\eta}\}} \frac{f(s)}{s+\sqrt{\eta}}\, ds\right]$$
$$= \frac{1}{2\pi i} \oint_{\{\pm\sqrt{\eta}\}} \frac{sf(s)}{s^2-\eta}\, ds$$

and

$$\beta_0 = \frac{1}{2\pi i} \oint_{\{\pm\sqrt{\eta}\}} \frac{f(s)}{s^2-\eta}\, ds, \qquad g(t) = \frac{1}{2\pi i} \oint_{\{t,\pm\sqrt{\eta}\}} \frac{f(s)}{(s-t)(s^2-\eta)}\, ds.$$

Upon substituting (3) in (1), we obtain

$$F_\eta(z) = z^{-1/3} \operatorname{Ai}(\eta z^{2/3})\alpha_0 - z^{-2/3} \operatorname{Ai}'(\eta z^{2/3})\beta_0$$
$$+ \frac{1}{2\pi i} \int_C e^{z[(1/3)t^3 - \eta t]}(t^2-\eta) g(t)\, dt.$$

An integration by parts gives

$$F_\eta(z) = z^{-1/3} \operatorname{Ai}(\eta z^{2/3})\alpha_0 - z^{-2/3} \operatorname{Ai}'(\eta z^{2/3})\beta_0$$
$$- \frac{1}{2\pi i} \int_C e^{z[(1/3)t^3 - \eta t]} f_1(t)\, dt,$$

where $f_1(t) = g'(t)$. Repeating this procedure we obtain the compound expansion

$$F_\eta(z) \sim z^{-1/3} \operatorname{Ai}(\eta z^{2/3}) \sum_{n=0}^{\infty} (-1)^n \frac{\alpha_n}{z^n}$$
$$- z^{-2/3} \operatorname{Ai}'(\eta z^{2/3}) \sum_{n=0}^{\infty} (-1)^n \frac{\beta_n}{z^n}, \tag{5}$$

where the coefficients α_n, β_n are defined as in (4) with the function f replaced with f_n, which in turn is defined by the scheme

$$f_{n+1}(t) = g_n'(t), \qquad f_n(t) = \alpha_n + \beta_n t + (t^2-\eta) g_n(t), \tag{6}$$

with $n = 0, 1, 2, \ldots$ and $f_0(t) = f(t)$. The expansion in (5) is valid for large values of z and holds uniformly with respect to η in a neighbourhood of the origin.

A more precise formulation can be given, but more information can be found in the literature; see [1] and [7].

The functions $f_n(t)$ defined in (6) can be represented in the form of Cauchy-type integrals. We have the following theorem.

Theorem 1. *Let the rational functions $R_n(s,t,\eta)$ be defined by*

$$R_0(s,t,\eta) = \frac{1}{s-t},$$
$$R_{n+1}(s,t,\eta) = \frac{-1}{s^2-\eta}\frac{d}{ds}R_n(s,t,\eta), \quad n = 0,1,2,\ldots, \tag{7}$$

where $s,t,\eta \in \mathbb{C}$, $s \neq t$, $s^2 \neq \eta$. Let $f_n(t)$ be defined by the recursive scheme (6), *where f_0 is a given analytic function in a domain G. Then we have*

$$f_n(t) = \frac{1}{2\pi i}\oint_{\mathcal{D}} R_n(s,t,\eta) f_0(s)\,ds,$$

where $\mathcal{D}$ is a simple closed contour in G that encircles the points t and $\pm\sqrt{\eta}$.

Proof. The proof starts with

$$f_n(t) = \frac{1}{2\pi i}\oint_{\mathcal{D}} R_0(s,t,\eta) f_n(s)\,ds,$$

and in this representation the recursion relation (6) for the functions f_n is used. More details can be found in [8]. □

For the coefficients α_n, β_n we have a similar representation:

$$\alpha_n = \frac{1}{2\pi i}\oint_{\mathcal{D}} A_n(s,\eta) f_0(s)\,ds, \qquad \beta_n = \frac{1}{2\pi i}\oint_{\mathcal{D}} B_n(s,\eta) f_0(s)\,ds, \tag{8}$$

where $\mathcal{D}$ is a simple closed contour in G that encircles the points $\pm\sqrt{\eta}$ and where $A_n(s,t)$ and $B_n(s,t)$ follow the same recursion (7) as the rational functions $R_n(s,t,\eta)$, with initial values

$$A_0(s,\eta) = \frac{s}{s^2-\eta}, \qquad B_0(s,\eta) = \frac{1}{s^2-\eta}.$$

We see that the coefficients α_n, β_n that play a role in the expansion (5) are well defined from an analytical point of view. However, from a computational point of view it may be quite difficult to evaluate the coefficients. For a simple rational function like $f_0(t) = 1/(t+1)$ the computations are rather straightforward, and we can even use residue calculus to evaluate the integrals in (8):

$$\alpha_n = -A_n(-1,\eta), \qquad \beta_n = -B_n(-1,\eta).$$

The first few values are in this case:

$$\alpha_0 = -\frac{1}{\eta - 1}, \qquad \beta_0 = \frac{1}{\eta - 1},$$
$$\alpha_1 = \frac{\eta + 1}{(\eta - 1)^3}, \qquad \beta_1 = -\frac{2}{(\eta - 1)^3},$$
$$\alpha_2 = -4\frac{2\eta + 1}{(\eta - 1)^5}, \qquad \beta_2 = 2\frac{\eta + 5}{(\eta - 1)^5},$$
$$\alpha_3 = 4\frac{2\eta^2 + 21\eta + 7}{(\eta - 1)^7}, \qquad \beta_3 = -40\frac{\eta + 2}{(\eta - 1)^7},$$
$$\alpha_4 = -280\frac{\eta^2 + 4\eta + 1}{(\eta - 1)^9}, \qquad \beta_4 = 40\frac{\eta^2 + 19\eta + 22}{(\eta - 1)^9},$$
$$\alpha_5 = 280\frac{\eta^3 + 29\eta^2 + 65\eta + 13}{(\eta - 1)^{11}}, \qquad \beta_5 = -1120\frac{2\eta^2 + 14\eta + 11}{(\eta - 1)^{11}}.$$

For a more complicated or general function $f_0(t)$ even computer algebra manipulations give complicated expressions which are very difficult to evaluate. In the next section we develop an algorithm for computing the coefficients α_n, β_n when the values of the derivatives of $f_0(t)$ at $t = \pm\sqrt{\eta}$ are available.

3. How to compute the coefficients α_n, β_n

We explain how the coefficients α_n, β_n of (5) can be computed. To avoid the square roots in the formulas we replace η with b^2, and we write (6) in the form

$$f_0(t) = f(t),$$
$$f_{n+1}(t) = g_n'(t), \qquad f_n(t) = \alpha_n + \beta_n t + (t^2 - b^2)g_n(t),$$

for $n = 0, 1, 2, \ldots$. We assume that the function f is analytic in a domain G, that the series expansions used in this section are convergent in G, and that the points $\pm b$ are inside G. Furthermore, we assume the coefficients $p_k^{(1)}$, $p_k^{(2)}$ of the expansions

$$f(t) = \sum_{k=0}^{\infty} p_k^{(1)}(t - b)^k, \qquad f(-t) = \sum_{k=0}^{\infty} p_k^{(2)}(t - b)^k \tag{9}$$

are available.

Theorem 2 (Algorithm). *Let coefficients f_k^e, f_k^o be defined by*

$$f_k^e = \frac{1}{2}\left[p_k^{(1)} + p_k^{(2)}\right], \quad f_k^o = \frac{1}{2}\left[p_k^{(1)} - p_k^{(2)}\right], \quad k = 0, 1, 2, \ldots,$$

and coefficients $f_k^{o,e}$ *by the recursion*

$$bf_k^{o,e} = f_k^o - f_{k-1}^{o,e}, \quad k \geqslant 0,$$

with $f_{-1}^{o,e} = 0$. *Next, define coefficients* γ_k, δ_k *by*

$$\gamma_0 = f_0^e, \qquad \delta_0 = f_0^{o,e},$$

and for $k \geqslant 1$:

$$\gamma_k = \sum_{j=1}^{k} \frac{(-1)^{k-j} j (2k-j-1)!}{(2b)^{2k-j} k! (k-j)!} f_j^e,$$

$$\delta_k = \sum_{j=1}^{k} \frac{(-1)^{k-j} j (2k-j-1)!}{(2b)^{2k-j} k! (k-j)!} f_j^{o,e}. \tag{10}$$

Finally, let for $n \geqslant 0$ *coefficients* $\gamma_k^{(n)}$, $\delta_k^{(n)}$ *be defined by the recursion*

$$\begin{aligned} \gamma_k^{(n+1)} &= (2k+1)\delta_{k+1}^{(n)} + 2b^2(k+1)\delta_{k+2}^{(n)}, \\ \delta_k^{(n+1)} &= 2(k+1)\gamma_{k+2}^{(n)}, \quad k = 0, 1, 2, \ldots, \end{aligned} \tag{11}$$

with $\gamma_k^{(0)} = \gamma_k$, $\delta_k^{(0)} = \delta_k$. *Then the coefficients* α_n, β_n *of expansion* (5) *are given by*

$$\alpha_n = \gamma_0^{(n)}, \quad \beta_n = \delta_0^{(n)}, \quad n \geqslant 0.$$

Proof. The coefficients f_k^e, f_k^o occur in the expansions

$$f_e(t) = \sum_{k=0}^{\infty} f_k^e (t-b)^k, \qquad f_o(t) = \sum_{k=0}^{\infty} f_k^o (t-b)^k,$$

where $f_e(t)$, $f_o(t)$ are the even and odd parts of f:

$$f_e(t) = \frac{1}{2}[f(t) + f(-t)], \qquad f_o = \frac{1}{2}[f(t) - f(-t)],$$

and the coefficients $f_k^{o,e}$ occur in the expansion

$$\frac{1}{t} f_o(t) = \sum_{k=0}^{\infty} f_k^{o,e} (t-b)^k.$$

The coefficients γ_k, δ_k occur in the expansion

$$f(t) = \sum_{k=0}^{\infty} \gamma_k (t^2 - b^2)^k + t \sum_{k=0}^{\infty} \delta_k (t^2 - b^2)^k. \tag{12}$$

Observe that

$$f_e(t)=\sum_{k=0}^{\infty}\gamma_k(t^2-b^2)^k, \qquad f_o(t)=t\sum_{k=0}^{\infty}\delta_k(t^2-b^2)^k,$$

and we will verify the first relation of (10). We write

$$\gamma_k=\frac{1}{2\pi i}\oint f_e\bigl(\sqrt{z+b^2}\bigr)\frac{dz}{z^{k+1}},$$

where the contour is a small circle around the origin. Also,

$$\gamma_k=\frac{1}{2\pi i}\oint f_e(t)\frac{2t\,dt}{(t+b)^{k+1}(t-b)^{k+1}},$$

where the contour is a small circle around $t=b$.

Substitute here the expansion $f_e(t)=\sum_{j=0}^{\infty}f_j^e(t-b)^j$. Then,

$$\gamma_k=\sum_{j=0}^{k}f_j^e\frac{1}{2\pi i}\oint\frac{2t\,dt}{(t+b)^{k+1}(t-b)^{k+1-j}}. \tag{13}$$

Expand

$$\frac{2t}{(t+b)^{k+1}}=\sum_{m=0}^{\infty}q_m(t-b)^m. \tag{14}$$

We find, by using [5, p. 108]

$$(1-z)^{-a}=\sum_{n=0}^{\infty}\frac{(a)_m}{m!}z^m=\sum_{n=0}^{\infty}\binom{-a}{m}(-z)^m,$$
$$q_m=(-1)^m\frac{(k-m)(k+m-1)!}{(2b)^{k+m}m!\,k!}.$$

When we use (14) in (13), we only need q_m with $m=k-j$. This gives the first result of (10). The proof for δ_k is the same, because $(1/t)f_o(t)$ is again even.

The coefficients $\gamma_k^{(n)}$, $\delta_k^{(n)}$ are used in

$$f_n(t)=\sum_{k=0}^{\infty}\gamma_k^{(n)}(t^2-b^2)^k+t\sum_{k=0}^{\infty}\delta_k^{(n)}(t^2-b^2)^k,$$

and the recursions in (11) are easily verified, as is the final relation

$$\alpha_n=\gamma_0^{(n)}, \quad \beta_n=\gamma_0^{(n)}, \quad n\geqslant 0. \qquad \square$$

The first few values are of the coefficients γ_k, δ_k of the expansion in (12) are

$$\gamma_0 = f_0^e, \qquad \delta_0 = \frac{1}{b} f_0^o,$$
$$\gamma_1 = \frac{1}{2b} f_1^e, \qquad \delta_1 = \frac{1}{2b^3}\left(b f_1^o - f_0^o\right),$$
$$\gamma_2 = \frac{1}{8b^3}\left(2b f_2^e - f_1^e\right), \qquad \delta_2 = \frac{1}{8b^5}\left(2b^2 f_2^o - 3b f_1^o + 3 f_0^o\right),$$

and we observe, as in (10), negative powers of b. From a computational point of view, this may cause numerical instabilities, because the coefficients are analytic functions of b at $b = 0$. For example, taking $f(t) = 1/(t+1)$ again, we obtain

$$\gamma_k = \frac{1}{(1-b^2)^{k+1}}, \quad \delta_k = -\frac{1}{(1-b^2)^{k+1}}, \quad k = 0, 1, 2, \ldots,$$

which follows from

$$\frac{1}{t+1} = \frac{1-t}{1-t^2} = \frac{1-t}{(1-b^2)-(t^2-b^2)}$$
$$= \sum_{k=0}^{\infty} \frac{(t^2-b^2)^k}{(1-b^2)^{k+1}} - t \sum_{k=0}^{\infty} \frac{(t^2-b^2)^k}{(1-b^2)^{k+1}}.$$

From the representations in, for example, (10), we conclude that if we apply the algorithm for computing the coefficients α_n, β_n of expansion (5), starting with numerical values of the coefficients $p_k^{(1)}$, $p_k^{(2)}$ of (9), we may encounter numerical instabilities when b is small. For this reason, it is important to use exact values of $p_k^{(1)}$, $p_k^{(2)}$, and computer algebra is of great help here. In the next section we consider a non-trivial case in which obtaining the exact values of the coefficients $p_k^{(1)}$, $p_k^{(2)}$ of (9) also needs special care.

Remark 1. In order to compute the coefficients α_n, β_n for $n = 0, 1, \ldots, N$ from the relation

$$\alpha_n = \gamma_0^{(n)}, \quad \beta_n = \delta_0^{(n)}, \quad n \geqslant 0,$$

and the recursion in (11), we need the starting values for this recursion γ_k, δ_k for $k = 0, 1, \ldots, 2N$. Hence, as follows from (10), we also need $p_k^{(1)}$, $p_k^{(2)}$, $k = 0, 1, \ldots, 2N$, in the expansions in (9).

4. Application to parabolic cylinder functions

Weber parabolic cylinder functions are solutions of the differential equation

$$\frac{d^2 y}{dx^2} - \left(\frac{1}{4}x^2 + a\right) y = 0. \tag{15}$$

Airy-type expansions for the solutions of this equation can be found in [2], and are obtained by using the differential equation. In this section we show how to obtain an integral representation like (1), and how to apply the algorithm of the previous section for deriving an Airy-type asymptotic expansion.

A standard solution of (15) is the integral

$$U(a,x)=\frac{e^{(1/4)x^2}}{i\sqrt{2\pi}}\int_C e^{-xs+(1/2)s^2}s^{-a-1/2}\,ds, \tag{16}$$

where the contour C is a vertical line in the complex plane with $\Re s>0$; see [4, p. 688, formula (19.5.4)].

We consider large negative values of a, and use Olver's notation

$$a=-\frac{1}{2}\mu^2, \qquad x=\mu t\sqrt{2}.$$

Changing the variable of integration by writing $s\to\mu s/\sqrt{2}$, we obtain

$$U\left(-\frac{1}{2}\mu^2,\mu t\sqrt{2}\right)=\frac{e^{(1/2)\mu^2t^2}}{i\sqrt{2\pi}}\left(\frac{\mu}{\sqrt{2}}\right)^{(1/2)\mu^2+1/2}\int_C e^{z\phi(s)}s^{-1/2}\,ds,$$

where

$$\phi(s)=\frac{1}{2}s^2-2st+\ln s, \qquad z=\frac{1}{2}\mu^2.$$

The saddle points are obtained from the equation $\phi'(s)=0$, that is, from $s^2-2st+1=0$, which gives two solutions

$$s_\pm=t\pm\sqrt{t^2-1}.$$

These points coalesce when $t\to\pm1$. Observe that in the new variables the differential equation (15) transforms into

$$\frac{d^2y}{dt^2}-\mu^4(t^2-1)y=0,$$

which has turning points at $t=\pm1$.

A transformation into the standard form (1) can be obtained by writing

$$\phi(s)=\frac{1}{3}w^3-\eta w+A, \tag{17}$$

where η and A have to be determined and do not depend on w. A transformation into the cubic polynomial is first considered in [9]. For further details on the theory of this method we refer to [1,7,8].

The parameters η and A are obtained by assuming that the saddle points $s_\pm$ in the s-variable should correspond with the saddle points $w_\pm=\pm\sqrt{\eta}$ in the w-variable. We write

$$t=\cosh\theta, \quad \text{which gives} \quad s_\pm=e^{\pm\theta}, \tag{18}$$

assuming for the time being that $\theta \geqslant 0$. We obtain the equations

$$\frac{1}{2}e^{+2\theta} - 2e^{+\theta}\cosh\theta + \theta = -\frac{2}{3}\eta^{3/2} + A,$$
$$\frac{1}{2}e^{-2\theta} - 2e^{-\theta}\cosh\theta - \theta = +\frac{2}{3}\eta^{3/2} + A,$$

from which we derive

$$\frac{4}{3}\eta^{3/2} = \sinh 2\theta - 2\theta, \qquad A = -\frac{1}{2} - \cosh^2\theta = -\frac{1}{2} - t^2. \tag{19}$$

By using these values of η and A the w-solution of the equation in (17) is uniquely defined. Namely we use that branch (of the three solutions) that is real for all positive values of s, and $s > 0$ corresponds with $w \in \mathbb{R}$.

After these preparations we obtain the standard form (cf. (1))

$$U\left(-\frac{1}{2}\mu^2, \mu t\sqrt{2}\right)e^{-zA} = \sqrt{2\pi}e^{(1/2)\mu^2 t^2}\left(\frac{\mu}{\sqrt{2}}\right)^{(1/2)\mu^2+1/2} F_\eta(z), \tag{20}$$

where

$$F_\eta(z) = \frac{1}{2\pi i}\int_C e^{z[(1/3)w^3-\eta w]} f(w)\,dw, \quad f(w) = \frac{1}{\sqrt{s}}\frac{ds}{dw}.$$

Taking into account the mapping in (17), we have

$$\frac{ds}{dw} = s\frac{w^2 - b^2}{s^2 - 2ts + 1}, \qquad f(w) = \sqrt{s}\frac{w^2 - b^2}{s^2 - 2ts + 1}, \qquad \eta = b^2. \tag{21}$$

As explained in the previous section, for the computation of the coefficients α_n, β_n, we need the coefficients $p_k^{(1)}$, $p_k^{(2)}$ of the expansions (cf. (9))

$$f(w) = \sum_{k=0}^{\infty} p_k^{(1)}(w-b)^k, \qquad f(-w) = \sum_{k=0}^{\infty} p_k^{(2)}(w-b)^k. \tag{22}$$

It turns out that $p_0^{(1)} = p_0^{(2)}$. Indeed, consider the expansions

$$s = s_+ + \sum_{k=1}^{\infty} s_k^+(w-b)^k, \qquad s = s_- + \sum_{k=1}^{\infty} s_k^-(w+b)^k. \tag{23}$$

Using the expression of ds/dw in (21) and l'Hôpital's rule we obtain

$$s_1^+ = s_+\frac{2b}{2(s_+ - t)s_1^+}, \quad \text{so that} \quad s_1^+ = \frac{\sqrt{bs_+}}{(t^2-1)^{1/4}} = \sqrt{\frac{bs_+}{\sinh\theta}}.$$

The square root has the plus sign because ds/dw is positive if $w \in \mathbb{R}$, as follows from the first relation in (21) and the properties of the mapping. From the expression (22) for $f(w)$ we obtain

$$p_0^{(1)} = f(b) = \frac{s_1^+}{\sqrt{s_+}} = \sqrt{\frac{b}{\sinh\theta}}.$$

Analogously,

$$s_1^- = \sqrt{\frac{bs_-}{\sinh\theta}} \quad\text{and}\quad p_0^{(2)} = f(-b) = \sqrt{\frac{b}{\sinh\theta}} = p_0^{(1)}. \tag{24}$$

In order to avoid expressions with algebraic roots in the computations, it is convenient to consider expansions like (22) for the function $\tilde{f}(w) = f(w)/f(b)$. We denote the corresponding coefficients by $\tilde{p}_k^{(1)}$ and $\tilde{p}_k^{(2)}$. Besides, to avoid algebraic roots in the expansions of (23), we replace t by a new variable

$$u = \sqrt{2b}\left(\frac{t-1}{t+1}\right)^{1/4}, \quad\text{so that}\quad t = \frac{4b^2+u^4}{4b^2-u^4}.$$

Then

$$s_+ = \frac{2b+u^2}{2b-u^2}, \qquad s_1^+ = \frac{2b+u^2}{2u}.$$

Other coefficients s_k^+ can be obtained by deriving a recurrence relation for them from the differential equation in (21). These are rational functions in u and b. The coefficients s_k^- can be obtained from the corresponding s_k^+ by changing the sign of both u and b. In particular,

$$s_- = \frac{2b-u^2}{2b+u^2}, \qquad s_1^- = \frac{2b-u^2}{2u}.$$

Further, the coefficients $\tilde{p}_k^{(1)}$ and $\tilde{p}_k^{(2)}$ can be computed using

$$f(w) = \frac{1}{\sqrt{s}}\frac{ds}{dw} = 2\frac{d\sqrt{s}}{dw}.$$

Recall that $\sqrt{s}$ satisfies the differential equation $2s\,dS/dw = S\,ds/dw$. It is convenient to compute the power series (in w–b) solution S_+ of this equation with $S_+(b) = 4u/(2b-u^2)$. Then $\tilde{f}(w) = dS_+/dw$ and the coefficients $\tilde{p}_k^{(1)}$ are obtained easily. The coefficients $\tilde{p}_k^{(2)}$ can be obtained by changing the sign of both b and u in the expression for $(-1)^k\tilde{p}_k^{(1)}$.

Application of the algorithm of the previous section gives the coefficients α_j, β_j for the expansion of $\tilde{f}(w)$, and these coefficients are rational functions in b and u. We write them in a more compact form as rational functions in $\eta = b^2$ and

$$\xi = \frac{u^4+4b^2}{4u^2} = \frac{bt}{\sqrt{t^2-1}}.$$

The first few coefficients in the expansion (5) are

$$\alpha_0 = 1, \qquad \beta_0 = 0,$$
$$\alpha_1 = \frac{1}{48}, \qquad \beta_1 = \frac{5\xi^3 - 6\eta\xi - 5}{48\eta^2},$$

$$\alpha_2 = \frac{385\xi^6 - 924\eta\xi^4 + 684\eta^2\xi^2 - 143\eta^3 + 70\xi^3 - 84\eta\xi - 455}{4608\eta^3},$$

$$\beta_2 = \frac{\beta_1}{48}, \qquad \alpha_3 = \frac{\alpha_2}{48} - \frac{2021}{34560}\alpha_1,$$

$$\beta_3 = \frac{425425\xi^9 - 1531530\eta\xi^8 + 2040012\eta^2\xi^5 - 28875\xi^6}{3317760\eta^5} + \frac{-1189005\eta^3\xi^3 + 69300\eta\xi^4 + 259110\eta^4\xi - 51300\eta^2\xi^2}{3317760\eta^5} + \frac{28875\xi^3 + 10725\eta^3 - 34650\eta\xi - 425425}{3317760\eta^5},$$

$$\beta_4 = \frac{\beta_3}{48} - \frac{2021}{34560}\beta_2.$$

The linear relations between the coefficients follow from expansion (8.11) in [2], see also [10]. Where both power series factors of Ai and Ai′ contain only even powers of our z (in Olver's notation, $z = (1/2)\mu^2$), but the whole expansion is multiplied by a function $g(z)$ with known asymptotics. Olver also notes that the coefficients in the Airy-type asymptotic expansion of $U(a, x)$ can be linearly determined from the asymptotic expansion (of the same function) in terms of elementary functions; see formulas (8.12), (8.13) in [2].

The coefficients α_n, β_n are analytic functions at $\eta = 0$ and we can expand them in Maclaurin series. The first few coefficients are expanded as follows:

$$\beta_1 = -\frac{9}{560} + \frac{7}{1800}\eta - \frac{1359}{1078000}\eta^2 + \frac{7}{16250}\eta^3 - \frac{152723}{1018710000}\eta^4 + \frac{3997}{75968750}\eta^5 + \cdots,$$

$$\alpha_2 = -\frac{199}{115200} + \frac{6849}{4928000}\eta - \frac{737}{1040000}\eta^2 + \frac{46711}{142560000}\eta^3 - \frac{975823}{6806800000}\eta^4 + \cdots.$$

The radius of convergence equals $(3\pi/2)^{2/3} = 2.81\ldots$. This number follows from the singularity of the mapping given in (19), with θ defined in (18). The mapping is singular at $t = -1$.

5. Maple code

As the input for the following code one has to (re)define functions AiryPw(k) and AiryPm(k) specifying the coefficients in (9). For example, if $f(t) = 1/(t+1)$, then one has to define

```
AiryPw := proc(k) (-1)^k/(1+AiryB)^(k+1) end:
AiryPm := proc(k) 1/(1-AiryB)^(k+1) end:
```

The output is given by functions AiryAlpha(n) and AiryBeta(n), which return the coefficients in (5). For convenience, one may rename the global variable AiryB using alias:

```
AiryAlpha := proc(n) normal(AiryGamma(n,0)) end:
AiryBeta := proc(n) normal(AiryDelta(n,0)) end:
AiryGamma := proc(n,k)
  if n=0 then AiryC(k)
  else factor((2*k+1)*AiryDelta(n-1,k+1)+2*AiryB^2*(k+1)*AiryDelta(n-1,k+2))
  fi
end:
AiryDelta := proc(n,k)
  if n=0 then AiryD(k)
  else 2*(k+1)*AiryGamma(n-1,k+2)
  fi
end:
AiryC := proc(k) local j;
  if k=0 then AiryFe(0)
  else factor(
    sum('(-1)^(k-j)*j/(2*k-j)*binomial(2*k-j,k)/(2*AiryB)^(2*k-j)*AiryFe(j)', 'j'=1..k))
  fi
end:
AiryD := proc(k) local j;
  if k=0 then AiryFoe(0)
  else factor(
    sum('(-1)^(k-j)*j/(2*k-j)*binomial(2*k-j,k)/(2*AiryB)^(2*k-j)*AiryFoe(j)', 'j'=1..k))
  fi
end:
AiryFoe := proc(k)
  if k<0 then 0
  else expand((AiryFo(k)-AiryFoe(k-1))/AiryB);
  fi
end:
AiryFe := proc(k) (AiryPw(k)+AiryPm(k))/2 end:
AiryFo := proc(k) (AiryPw(k)-AiryPm(k))/2 end:
```

The remaining code computes the coefficients of the expansion of parabolic cylinder function $U(a, x)$ (we use the normalized function $f(w)/f(b)$, as mentioned after (24)). To use the code one has to assign

```
AiryPw := ParCyPw;      AiryPm := ParCyPm;
```

The global variables are AiryB, ParCyU, ParCyXi, they correspond to variables b, u, ξ in the text. The coefficients in b and u would be returned by AiryAlpha(n) and AiryBeta(n), and coefficients in b and ξ by ParCyAlpha(n) and ParCyBeta(n):

```
alias(ParCyUa=RootOf(z^4-4*ParCyXi*z^2+4*AiryB^2,z)):
# Algebraic relation between ParCyU and ParCyXi
ParCyAlpha := proc(k) factor(evala(subs(ParCyU=ParCyUa,AiryAlpha(k)))) end:
ParCyBeta := proc(k) factor(evala(subs(ParCyU=ParCyUa,AiryBeta(k)))) end:
```

```
ParCySw := proc(k) option remember; local T, s, a, w;
  if k=0 then (2*AiryB+ParCyU^2)/(2*AiryB-ParCyU^2)
  elif k=1 then (2*AiryB+ParCyU^2)/2/ParCyU
  else s := sum('a[i]*w^i', 'i'=0..k);
    T := coeff(expand((s^2+2*(ParCyU^4+4*AiryB^2)/(ParCyU^4-4*AiryB^2)*s+1)
       *diff(s,w)-w*(w+2*AiryB)*s), w, k);
    sort(factor(solve(subs(seq(a[i]=ParCySw(i), i=0..k-1), T), a[k])), ParCyU)
  fi
end:
ParCySm := proc(k) subs(ParCyU= -ParCyU, AiryB= -AiryB, ParCySw(k)) end:
ParCySqrtS := proc(k) option remember; local j;
  if k=0 then 4*ParCyU/(2*AiryB-ParCyU^2)
  else factor(sum('(3/2*j-k)*ParCySw(j)*ParCySqrtS(k-j)', 'j'=1..k)/ParCySw(0)/k)
  fi end:
ParCyPw := proc(k) (k+1)*ParCySqrtS(k+1) end:
ParCyPm := proc(k) (-1)^k*subs(AiryB= -AiryB, ParCyU= -ParCyU, ParCyPw(k)) end:
```

Acknowledgment

The authors thank the referee for valuable comments.

References

[1] F.W.J. Olver, Asymptotics and Special Functions, Academic Press, New York, 1974, reprinted in 1997 by A.K. Peters.
[2] F.W.J. Olver, Uniform asymptotic expansions for Weber parabolic cylinder functions of large orders, J. Res. NBS 63B (1959) 131–169.
[3] T.M. Dunster, Uniform asymptotic expansions for Whittaker's confluent hypergeometric functions, SIAM J. Math. Anal. 20 (1989) 744–760.
[4] M. Abramowitz, I. Stegun, Handbook of Mathematical Functions with Formulas, Graphs and Mathematical Tables, Nat. Bur. Standards Appl. Ser., Vol. 55, U.S. Government Printing Office, Washington, DC, 1964, paperback edition published by Dover, New York.
[5] N.M. Temme, Special Functions: An Introduction to the Classical Functions of Mathematical Physics, Wiley, New York, 1996.
[6] N. Bleistein, Uniform asymptotic expansions of integrals with stationary points and algebraic singularity, Comm. Pure Appl. Math. 19 (1966) 353–370.
[7] R. Wong, Asymptotic Approximations of Integrals, Academic Press, New York, 1989.
[8] A.B. Olde Daalhuis, N.M. Temme, Uniform Airy type expansions of integrals, SIAM J. Math. Anal. 25 (1994) 304–321.
[9] C. Chester, B. Friedman, F. Ursell, An extension of the method of steepest descent, Proc. Cambridge Philos. Soc. 53 (1957) 599–611.
[10] N.M. Temme, Numerical and asymptotic aspects of parabolic cylinder functions, J. Comput. Appl. Math. 121 (2000) 221–246.